
Excepto quando explicitamente impedido, podes usar predicados / procedimentos feitos nas aulas ou existentes no Prolog, sem os definir, por exemplo member/2 e findall/3.
Considera um fragmento de uma base de conhecimentos com predicados como os seguintes:

% rock_band(Band, Members)

rock_band(superheavy, [mick, dave, joss, damian, a_r_rahman]).

rock_band(guns_n_roses, [axl, slash, izzy, duff, steven]). 

rock_band(gnr, [toli, jorge, rui, ruka, andrew, hugo]). 

% cd(Band, CDTitle)

cd(superheavy, superheavy).

cd(guns_n_roses, chinese_democracy).

cd(gnr, retropolitana).

cd(gnr, popless).
I – Prolog Declarativo
Neste grupo, não se podem usar ações, nem mecanismos de controlo, nem meta‑predicados.

(2.5 Valores) 1 – Define o predicado rock_band_member/2, tal que rock_band_member(Band, Member) significa que Member é um elemento da banda Band. Exemplo:
?- rock_band_member(guns_n_roses, slash).

yes

?- rock_band_member(guns_n_roses, X).

X = axl;

X = slash

?- rock_band_member(B, rui).

B = gnr

Não faças validações dos argumentos dos predicados.
R:

rock_band_member(Band, Member):-

rock_band(Band, Elements),

member(Member, Elements).

(2.5 Valores) 2 – Define o predicado rock_bands/2 tal que rock_bands(Bands, ListOfBandsMembers) significa que ListOfBandsMembers é a lista de listas de elementos das bandas de Bands. Exemplo:

?- rock_bands([guns_and_roses, gnr], List).

List = [[axl, slash, izzy, duff, steven], [toli, jorge, rui, ruka, andrew, hugo]]

yes
Não faças validações dos argumentos dos predicados.

R:

rock_bands([B|Bands], [Musicians|Members]):-


rock_band(B, Musicians),


rock_bands(Bands, Members).

rock_bands([], []).
(2.5 Valores) 3 – Define o predicado largest_band/2 tal que largest_band(Bands, LargestBand) significa que LargestBand é uma das banda com o maior número de elementos de todas as bandas da lista Bands. Exemplo:

?- largest_band([guns_and_roses, gnr], Band).

Band = gnr
yes
Garante que o primeiro argumento é uma lista de bandas e que o segundo é uma variável ou o nome de uma banda.

NOTA: Neste exercício não podes usar o perdicado lenght/2 do Prolog. Se necessitares saber o comprimento de uma lista, tens de definir um predicado para isso.

R:

largest_band(Bands, LargestBand):-


is_list(Bands),


(var(LargestBand); atom(LargestBand)),


aux_largest_band(Bands, LargestBand).
aux_largest_band([B1, B2 | Bands], LargestBand):-


aux_largest_band([B2 | Bands], Large),

larger(B1, Large, LargestBand).

aux_largest_band([Band], Band).
% ESTA PARTE DAQUI EM DIANTE ESTÁ FEITA DE MODO A SER MAI EFICIENTE

% DEPOIS APRESENTO UMA VERSÃO MAIS SIMPLES

larger(Band1, Band2, Larger):-


bandsize(Band1, Size1),


bandsize(Band2, Size2),


larger(Band1, Size1, Band2, Size, Larger).

larger(Band1, Size1, _, Size2, Band1):-


Size1 >= Size2.

larger(_, Size1, Band2, Size2, Band2):-


Size2 > Size1.
bandsize(Band, Size):-

rock_band(Band, Musicians),

listsize(Musicians, Size).
listsize([X|Rest], Size):-


listsize(Rest, S),


Size is S + 1.

listsize([], 0).
% VERSÃO MAIS SIMPLES

larger(Band1, Band2, Band1):-


bandsize(Band1, Size1),


bandsize(Band2, Size2),


Size1 >= Size2.

larger(Band1, Band2, Band2):-


bandsize(Band1, Size1),


bandsize(Band2, Size2),


Size2 > Size1.

(2.5 Valores) 4 – Define o predicado musicians_bands/2 tal que musicians_bands(Musicians, Bands) significa que Bands é a menor lista bandas que inclui todos os músicos da lista Musicians. Exemplo:

?- musicians_bands([rui, ruka, mick], Bands).

Bands = [gnr, superheavy];

no
?- musicians_bands([rui, peter], Bands).

no

Garante que o primeiro argumento seja uma lista de nomes de músicos, e que o segundo argumento seja uma variável ou uma lista de nomes de bandas. A ordem pela qual as bandas aparecem na lista é irrelevante. Se necessário, admite que rock_band_member/2 está já implementado, mesmo que não tenhas respondido à pergunta 1.
R:

musicians_bands(Musicians, Bands):-


is_list(Musicians),


(var(Bands); is_list(Bands)), % list/1, no LPA


musicians_bands(Musicians, [], Bands).

% ESTA PARTE DAQUI EM DIANTE ESTÁ FEITA DE MODO A SER MAI EFICIENTE

% DEPOIS APRESENTO UMA VERSÃO MAIS SIMPLES

musicians_bands([M|Musicians], Ac, Bands):-


rock_band_member(Band, M),


musicians_bands(Musicians, Band, Ac, Bands).

musicians_bands([], Ac, Ac).

musicians_bands([M|Musicians], Band, Ac, Bands):-


member(Band, Ac),


musicians_bands([M|Musicians], Ac, Bands).

musicians_bands([M|Musicians], Band, Ac, Bands):-


\+ member(Band, Ac),

musicians_bands([M|Musicians], [Band|Ac], Bands).

musicians_bands([], Band, Ac, Ac):-


member(Band, Ac).
musicians_bands([], Band, Ac, [Band|Ac]):-

\+ member(Band, Ac).
% VERSÃO MAIS SIMPLES

musicians_bands([M|Musicians], Ac, Bands):-


rock_band_member(Band, M),


member(Band, Ac),
musicians_bands(Musicians, Ac, Bands).

musicians_bands([M|Musicians], Ac, Bands):-


rock_band_member(Band, M),


\+ member(Band, Ac),
musicians_bands(Musicians, [Band | Ac], Bands).

musicians_bands([], Ac, Ac).

II – Prolog Procedimental e Input/Output
Neste grupo, podem e devem usar-se ações, mecanismos de controlo, e meta‑predicados.

(2.5 Valores) 5 – Define o predicado rock_band_cds/2 tal que rock_band_cds(Band, CDs) significa que CDs é a lista dos cds da banda Band. Exemplo:

?- rock_band_cds(Band, L).

Band = superheavy

L = [superheavy];

Band = guns_n_roses

L = [chinese_democracy];

Band = gnr

L = [retropolitana, popless]

yes

Não faças validações dos argumentos dos predicados.

R:

rock_band_cds(Band, CDs):-


bagof(Cd, cd(Band, Cd), CDs).

(2.5 Valores) 6 – Sem criar uma lista de todas as bandas, escreve o procedimento nBands/1 que calcula o número de bandas de rock existentes. Exemplo
?- nBands(N).

N = 3

yes

Não faças validações do argumento do procedimento.

R:

nBands(N):-

assert(count(0)),

count_bands,

retract(count(N)).

count_bands :-

rock_band(_, _),

update_counter,

fail.

count_bands.

update_counter:-


retract(count(N)),


N1 is N + 1,


assert(count(N1)),


!.

(2.5 Valores) 7 - Escreve o procedimento band_elements/0 que escreve, no ecrã, os nomes dos elementos das bandas cujos nomes são dados pelo utilizador. A repetição termina quando o utilizador introduzir o número -1, em vez do nome de uma banda.
Exemplo:
?- band_elements.

Band (enter -1 to exit): gnr. <ENTER>

toli, jorge, rui, ruka, andrew and hugo.
Band (enter -1 to exit): guns_and_roses. <ENTER>

axl, slash, izzy, duff and steven

Band (enter -1 to exit): -1. <ENTER>

yes

Admite que existe o procedimento write_musicians/1 que escreve uma lista de músicos, separados por vírgulas, excepto o último que é separado pela conjunção “and” (se adequado).
Na leitura dos nomes das bandas, não é necessário fazer validações.

R:

band_elements :-

repeat,



read_band_name(Band),


     
band_elements(Band), !.

read_band_name(Band):-

write(‘Band (enter -1 to exit): ‘),


read(Band).

band_elements(-1):- !.

band_elements(Band):-


rock_band(Band, Musicians),


write_musicians(Musicians),


!,


fail.

% Não faz parte da resolução (foi oferecido)

write_musicians([M1, M2, M3 | Musicians]):-


write(M1), write(‘, ‘),


write_musicians([M2, M3 | Musicians]).

write_musicians([M1, M2]):-


write(M1), write(‘ and ‘), write(M2), nl.

write_musicians([M]):- write(M), nl.

(2.5 Valores) 8 – Imagina que o sistema pretende obter os nomes completos das bandas e dos músicos. Nesse contexto, poderia existir o procedimento musician_name/2 que apresenta ao utilizador o nome de um músico, conforme ele é conhecido pelo programa (e.g., mick) e pede para o utilizador indicar o nome completo do músico (e.g., Mick Jagger).

Exemplo:
?- musician_name(mick, Name).

What is the name of musician mick? Mick Jagger<ENTER>

Name = ‘Mick Jagger’
yes

Usando a leitura caracter a caracter, mas sem usar o predicado já feito na aula, escreve o procedimento musician_name/2 acabado de descrever. Não são necessárias validações.

R:

musician_name(InternalName, TrueName) :-

write(‘What is the name of musician ‘),

write(InternalName),

write(‘? ‘),


readname(TrueName).
readname(Name):-


get0(Ch),

getname(Ch, NameChars),

atom_codes(Name, NameChars). % SWI. No LPA, atom_chars/2
getname(Ch, []):-

end_of_name(Ch), !.

getname(Ch, [Ch | Rest]):-


get0(Ch1),


getname(Ch1, Rest).
end_of_name(10). % SWI. No LPA é 13


