
Teste de Prolog

Grande parte dos enunciados das perguntas deste teste relaciona-se com um programa para resolver exercícios de Sudoku. Um exercício de Sudoku é representado por uma lista de linhas, por exemplo:
[[9, 4, +, 1, +, 2, +, 5, 8],
 [6, +, +, +, 5, +, +, +, 4],
 [+, 2, +, +, +, +, +, 6, +],
 [5, +, 8, +, 2, +, 4, +, 1],
 [+, 6, +, +, +, +, +, 8, +],
……………………
]
As posições não preenchidas são representadas por + mas isso não deve interferir na resolução de nenhuma das perguntas do teste.
I – Prolog Declarativo
Neste grupo não podem ser usados os mecanismos de controlo “cut”, fail e repeat, nem os meta-predicados assert/1, retract/1, findall/3, bagof/3 e setof/3.
(4 Valores) 1 -	Escreve o predicado nth/3, tal que nth(N, L, X) significa que X é o elemento de ordem N da lista L.
Exemplos
?- nth(2, [a, b, c, d], X).
X = b
?- nth(10, [a, b, c], X).
false (há interpretadores de Prolog que, em vez de false, imprimem no)
R:
nth(1, [X|_], X).
nth(N, [_|L], Y):-
	N > 1,
	N1 is N - 1,
	nth(N1, L, Y).
[bookmark: aqui](4 Valores) 2 -	Usando o predicado nth/3 da pergunta 1, quer tenhas respondido corretamente quer não, define o predicado nth_column/3, tal que nth_column(Rows, N, Column) significa que Column é a coluna formada pelos elementos de ordem N das linhas de Rows.
Exemplos
?- nth_column([[a, b, c], [d, e, f], [g, h, i]], 2, C).
C = [b, e, h]
?- nth_column([[a, b, c], [d, e, f], [g, h, i]], 3, C).
C = [c, f, i]
?- nth_column([[a, b, c], [d, e, f], [g, h, i]], 0, C).
false (há interpretadores de Prolog que, em vez de false, imprimem no)
Não faças validações dos argumentos passados na chamada de nth_column/3.
R:
nth_column([], _, []).
nth_column([R|Rows], N, [X|Column]):-
	nth(N, R, X),
	nth_column(Rows, N, Column).
(4 Valores) 3 -	Escreve o predicado hipoteses_possiveis/2, tal que hipoteses_possiveis(L1, L2) significa que L2 é a lista de inteiros de 1 a 9 que não fazem parte de L1.
Exemplo
?- hipoteses_possiveis([1, 4, 7, 3, 2, 6], Hipoteses).
Hipoteses = [5, 8, 9]
Não faças validações dos argumentos do predicado.
Sugestão: Embora não sendo obrigatório, sugiro que consideres a seguinte abordagem inicial
hipoteses_possiveis(L1, L2):- hipoteses_possiveis(1, 9, L1, L2).
No predicado hipoteses_possiveis/4, o primeiro argumento é um contador iniciado com 0; e o segundo argumento é o valor a que o contador deve chegar (9, neste caso).
R:
hipoteses_possiveis(I, N, L1, [I|L2]):-
	I < N,
	\+ member(I, L1),
	I1 is I + 1,
	hipoteses_possiveis(I1, N, L1, L2).
hipoteses_possiveis(I, N, L1, L2):-
	I < N,
	member(I, L1),
	I1 is I + 1,
	hipoteses_possiveis(I1, N, L1, L2).
hipoteses_possiveis(N, N, L, [N]):-
	\+ member(N, L).
hipoteses_possiveis(N, N, L, []):-
	member(N, L).
II – Prolog Procedimental
(4 Valores) 4 -	Admite que tens o predicado sudoku_solution/2, tal que sudoku_solution(Problem, Solution) significa que Solution é uma solução para o problema de Sudoku representado em Problem. O problema e a solução representam-se da mesma forma, como explicado no início do enunciado.
[bookmark: _GoBack]Admitindo que podem existir várias soluções para cada problema, escreve o procedimento repetitivo, não recursivo, print_solutions/1 que recebe um problema de sudoku e mostra, no ecrã do computador, todas as soluções válidas de um problema. Para este procedimento, usa o predicado valid_problem/1 (sem o definir), que verifica que Problem é uma representação correta de um problema de Sudoku e o predicado display_sudoku /1 (sem o definir), que pode imprimir tanto o problema como a solução.
R:
print_solutions(Problem):-
	valid_problem(Problem),
	sudoku_solution(Problem, Solution),
	display_solution(Solution),
	fail.
print_solutions(_).
(4 Valores) 5 -	Cada problema de sudoku é um termo Prolog (uma lista de 9 listas de 9 elementos cada), armazenado num ficheiro. Cada ficheiro tem um único problema.
Escreve o predicado play_sudoku/0 que pede ao utilizador o pathname do ficheiro com o problema, abre e lê o problema, imprime todas as soluções no ecrã, e volta a pedir um ficheiro com um novo problema ao utilizador. Quando o utilizador não pretende jogar mais, introduz a palavra end, em vez do pathname do ficheiro.
Para este exercício, podes usar o predicado print_solutions/1, da pergunta 5, mesmo que não tenhas respondido corretamente a essa pergunta.
Não faças nenhuma validação, nem na interação com o utilizador nem na leitura do problema do ficheiro.
R:
play_sudoku :-
   seen, % Precaução contra ficheiros abertos. Não é obrigatório na solução
   repeat,
     read_input_filename(File),
     process_user_input(File), !.
read_input_filename(FileName):-
	write('Sudoku file or "end" to finish: '),
	read(FileName).
process_user_input(end):- !.
process_user_input(File):-
	see(File),
	read(SudokuProblem),
	print_solutions(SudokuProblem),
	seen,
	!,
	fail.

